
1 Preliminaries

1.1 Introduction

SOFA stands for Standards Of Fundamental Astronomy. The SOFA software is a collection
of Fortran 77 and C subprograms that implement official IAU algorithms for fundamental-
astronomy computations. At the present time the SOFA software comprises 131 astronomy
routines supported by 55 utility (mainly vector/matrix) routines. The core documentation for
the SOFA collection consists of classified and alphabetic lists of subroutine calls plus detailed
preamble comments in the source code of individual routines.

The present document looks at a selection of SOFA routines that deal with dates and times. It
provides a tutorial introduction to timekeeping and the main time scales and calendars used by
astronomers. Short examples demonstrate how to call SOFA routines to perform the sorts of
conversions and transformations that may be needed in applications where times and dates are
involved.

1.2 Quick start

Readers already familiar with the elementary concepts can safely omit the explanatory material
and refer directly to the examples in Sections 1.5 and 4.4, and others that can be found embedded
in the text. Figure 2 shows the supported time scales and the names of the transformation
routines.

1.3 The SOFA time and date routines

The SOFA routines to be discussed are the following:

CAL2JD Gregorian calendar to Julian day number
D2TF decompose days into hours, minutes, seconds
D2DTF decode internal format into time and date numbers
DAT ∆AT (= TAI−UTC) for a given UTC date
DTDB TDB−TT
DTF2D encode time and date numbers into internal format
EPB Julian date to Besselian epoch
EPB2JD Besselian epoch to Julian date
EPJ Julian date to Julian epoch
EPJ2JD Julian epoch to Julian date
JD2CAL Julian date to Gregorian year, month, day, fraction
JDCALF Julian date to Gregorian date for formatted output
TAITT TAI to TT
TAIUT1 TAI to UT1
TAIUTC TAI to UTC
TCBTDB TCB to TDB

TCGTT TCG to TT
TDBTCB TDB to TCB
TDBTT TDB to TT
TF2D hours, minutes, seconds into days
TTTAI TT to TAI
TTTCG TT to TCG
TTTDB TT to TDB
TTUT1 TT to UT1
UT1TAI UT1 to TAI
UT1TT UT1 to TT
UT1UTC UT1 to UTC
UTCTAI UTC to TAI
UTCUT1 UTC to UT1

1.4 Intended audience

The SOFA time routines are designed for convenient practical use, with levels of rigor that are
consistent with this goal. Inevitably, many fine details are glossed over, such as transforming
topocentric proper time to TAI, different realizations of TT, competing TDB−TT models, and so
on. Furthermore, the “proleptic” issue is largely ignored; for instance the routines will blithely
perform a transformation from TAI to TT for a date in the nineteenth century, long before
either time scale was introduced. These simplifications notwithstanding, time specialists may
nevertheless find SOFA useful as a reliable source of comparison results.

1.5 A simple example: UTC to TT

A particularly common application is when a moment in one time scale is to be referred to
another time scale. This requires three steps:

1. Call the DTF2D routine to transform the date and time into the SOFA internal format.

2. Call the appropriate sequence of transformation routines (see Figure 2).

3. Call the D2DTF routine to prepare the transformed time for output.

For example, to transform 2010 July 24, 11:18:07.318 (UTC) into terrestrial time TT, and report
it, rounded to 1 ms precision:

INTEGER J, IY, IM, ID, IHMSF(4)

DOUBLE PRECISION U1, U2, A1, A2, T1, T2

* Encode UTC date and time into internal format.

CALL iau_DTF2D ( ’UTC’, 2010, 7, 24, 11, 18, 7.318D0,

: U1, U2, J )

IF ( J.NE.0 ) STOP

* Transform UTC to TAI, then TAI to TT.

CALL iau_UTCTAI ( U1, U2, A1, A2, J )

IF ( J.NE.0 ) STOP

CALL iau_TAITT ( A1, A2, T1, T2, J )

IF ( J.NE.0 ) STOP

* Decode and report the TT.

CALL iau_D2DTF ( ’TT’, 3, T1, T2, IY, IM, ID, IHMSF, J )

IF ( J.NE.0 ) STOP

WRITE ( *, ’(I5,2(’’/’’,I2.2),I3,2(’’:’’,I2.2),’’.’’,I3.3)’)

: IY, IM, ID, IHMSF

The output is “2010/07/24 11:19:13.502”.

1.6 Abbreviations

GPS Global Positioning System
GR General relativity
IAU International Astronomical Union
IERS International Earth Rotation and reference systems Service
JD Julian date
J2000.0 2000 January 1.5 (in some specified time scale)
MJD Modified Julian date
SOFA Standards of Fundamental Astronomy
TAI International Atomic Time
TCB Barycentric Coordinate Time
TCG Geocentric Coordinate Time
TDB Barycentric Dynamical Time
TT Terrestrial Time
UT,UT1 Universal Time
UTC Coordinated Universal Time

2 Times and dates

2.1 Timekeeping basics

Timekeeping means following an agreed recipe for measuring time, using some natural “clock”
as a basis. The most practical phenomena for this purpose are rotations and oscillations, and
for most of history the Earth’s rotation was the best available timekeeper. Because time was
synonymous with the cycle of day and night, the units used to express time intervals reflect that
choice: days, divided into hours, minutes and seconds. For larger intervals, other astronomical
phenomena play a role, in particular the orbital periods of the Earth and Moon.

Repeated attempts in the 19th century to model the motion of solar system objects, especially
the fast-moving Moon, were only partially successful. Each new theory, while reproducing
existing data accurately, would soon start to diverge from observation. The pragmatic solution
was to create a tautology by regarding the theories as clocks in their own right; in effect the
clock’s hands were the bodies of the solar system, reading out “ephemeris time”. However, once
man-made clocks became sufficiently accurate, suspicions that irregularities in Earth rotation
had all along been the limiting factor were confirmed, and laboratory time scales created by such
clocks—nowadays atomic, before that quartz—took over as the primary timekeeping standard.

2.2 Formatting conventions

A consequence of the history of timekeeping (and, in the case of the calendar, the nature of
the problem) is that the rules for expressing and transforming times and dates are somewhat
complicated and inconvenient.

To convert a time interval between hours, minutes, seconds and days, SOFA provides two rou-
tines: D2TF breaks an interval into hours, minutes and seconds while TF2D does the reverse.
Here is an example, where a time is expressed as a fraction of a day; a moment 0d.25 earlier is
then expressed in hours, minutes and seconds, to a resolution of 1 ms:

CHARACTER PM

INTEGER IH, IM, J, IHMSF(4)

DOUBLE PRECISION S, F

* The time.

IH = 23

IM = 5

S = 11.630799D0

WRITE ( *, ’(I3,’’:’’,I2.2,’’:’’,F9.6)’ ) IH, IM, S

* Express as a fraction of 1 day.

CALL iau_TF2D ( ’+’, IH, IM, S, F, J )

IF ( J.NE.0 ) STOP

WRITE ( *, ’(F15.12)’ ) F

* Six hours earlier.

F = F - 0.25D0

* Report to 1 ms precision.

CALL iau_D2TF ( 3, F, PM, IHMSF )

WRITE ( *, ’(1X,I2,2(’’:’’,I2.2),’’.’’,I3.3)’ ) IHMSF

The output (i.e. test time, equivalent fraction of a day, time six hours earlier) is:

23:05:11.630799

0.961940171285

17:05:11.631

The familiar Gregorian calendar date, consisting of year, month and day, is designed to keep
more or less in step with the tropical year, the year of the seasons, which has a length of about
365.2422 days. Its predecessor, the Julian calendar, approximated the tropical year by using a
basic 365-day year and introducing an extra day (February 29) in every fourth year, giving an
average year length of 365.25 days. Astronomers still use the latter, the terms Julian year and
Julian century meaning exactly 365.25 and 36525 days respectively. The Gregorian calendar
provided a better approximation to the tropical year by dropping three such leap years in each
400 years, to give an average of 365.2425 days. The rule is that century years must be divisible
by 400 to be a leap year; other years are leap years if they are divisible by 4. So the year 2000
was a leap year, but 2100 will not be a leap year.

2.3 Julian date

For many purposes, calendar date is inconvenient: what is needed is a continuous count of
days. For this purpose the system of Julian day number can be used. JD zero is located about
7000 years ago, well before the historical era, and is formally defined in terms of Greenwich
noon;1 for example Julian day number 2449444 began at noon on 1994 April 1.

Julian date (JD) is the same system but with a fractional part appended; JD 2449443.5 was the
midnight on which 1994 April 1 commenced. Because of the unwieldy size of Julian Dates and
the awkwardness of the half-day offset, it is accepted practice to remove the leading ‘24’ and the
trailing ‘.5’, producing what is called the Modified Julian Date:

MJD = JD− 2400000.5.

Thus 1994 April 1 commenced at MJD 49443.0.

1Use of JD and MJD is not restricted to “Greenwich time” and can be used in conjunction with other time
scales (Section 3.2) such as TAI, TT and TDB. However, UTC poses problems because of the ambiguity that
occurs during a leap second (Section 3.5.1).

MJD is often used in computer applications, rather than JD itself, to reduce exposure to rounding
errors. SOFA goes one step further, by always expressing JD as two double precision numbers,
the sum of which is the desired JD. The user decides how best to apportion the JD. In applica-
tions where precision is not critical, one of the two parts can simply be set to zero and the other
to the JD itself. A better compromise might be 2400000.5 as one of the numbers, and the MJD
as the other. The various SOFA routines go to some trouble to use the two parts wisely—they
are not simply added together—and the preamble comments in the various routines indicate
which split will give optimal accuracy.

The SOFA routine CAL2JD expresses a given Gregorian calendar date as a two-part JD. The
routine JD2CAL takes a two-part JD and breaks it into Gregorian calendar date plus a fraction
of a day. The routine JDCALF does a similar job, but rounds to a specified precision and returns
the fraction as an integer, ready for use in a message or report. Here is a demonstration, using
a deliberately awkward date and time close to the end of a month:

INTEGER IY, IM, ID, IHOUR, IMIN, J, IYMDF(4)

DOUBLE PRECISION D1, D2, SEC, D, FD

* Date and time.

IY = 2008

IM = 2

ID = 29

IHOUR = 23

IMIN = 59

SEC = 59.9D0

WRITE ( *,

: ’(I5,2(’’/’’,I2.2),I3.2,’’:’’,I2.2,’’:’’,F4.1 )’ )

: IY, IM, ID, IHOUR, IMIN, SEC

* Express as 2-part JD.

CALL iau_CAL2JD ( IY, IM, ID, D1, D2, J )

IF ( J.NE.0 ) STOP

CALL iau_TF2D ( ’+’, IHOUR, IMIN, SEC, D, J )

IF ( J.NE.0 ) STOP

D2 = D2+D

WRITE ( *, ’(F10.1,’’ +’’,F13.6,’’ =’’,F15.6)’ )

: D1, D2, D1+D2

* Express as calendar date and fraction of a day.

CALL iau_JD2CAL ( D1, D2, IY, IM, ID, FD, J )

D = DBLE(ID)+FD

WRITE ( *, ’(I5,’’/’’,I2.2,’’/’’,F9.6)’ ) IY, IM, D

IF ( J.NE.0 ) STOP

* Round to 0.001 day.

CALL iau_JDCALF ( 3, D1, D2, IYMDF, J )

IF ( J.NE.0 ) STOP

WRITE ( *, ’(I5,2(’’/’’,I2.2),’’.’’,I3.3)’ ) IYMDF

The output (i.e. date and time, MJD and JD equivalents, date with fractional day, the same
but rounded) is:

2008/02/29 23:59:59.9

2400000.5 + 54525.999999 = 2454526.499999

2008/02/29.999999

2008/03/01.000

2.4 Besselian and Julian epochs

For some astronomical purposes it is convenient to work in fractional years, such that a given
date and time near the end of 2009 (for example) can be written “2009.93”.

Formerly, this was done using a system called Besselian epoch. The unit is tropical years (about
365.2422 days), the time scale is ephemeris time (Section 3.2) and the Besselian year begins
when the ecliptic longitude of the mean Sun is 280◦—which occurs near the start of the calendar
year and which is why that particular round figure was chosen. Besselian epochs are mainly
associated with older star catalogues, specifying the mean equator and equinox and hence the
celestial orientation of the equatorial coordinate system used by the catalogue concerned.

Julian epoch took over from the beginning of 1984. It uses the Julian year of exactly 365.25 days,
and the TT time scale (Section 3.2); Julian epoch 2000 is defined to be 2000 January 1.5, which
is JD 2451545.0 or MJD 51544.5.

The two types of epoch are denoted by a prefix ‘B’ or ‘J’, hence “B1950.0” and “J2000.0”. In
the absence of such a prefix, it can be assumed that epochs before 1984.0 are Besselian and, from
1984.0 on, Julian. The transformations between MJD and the two types of epoch are carried
out by the SOFA routines EPB, EPB2JD, EPJ and EPJ2JD. They are demonstrated in the
following code:

DOUBLE PRECISION D, E, D1, D2, iau_EPB, iau_EPJ

* Julian Date.

D = 2457073.05631D0

WRITE ( *, ’(F14.5)’ ) D

* Transform into Besselian epoch.

E = iau_EPB ( 0D0, D )

WRITE ( *, ’(’’ B’’,F15.10)’ ) E

* Transform back.

CALL iau_EPB2JD ( E, D1, D2 )

WRITE ( *, ’(F18.9)’ ) D1+D2

* The same for Julian epoch.

E = iau_EPJ ( 0D0, D )

WRITE ( *, ’(’’ J’’,F15.10)’ ) E

CALL iau_EPJ2JD ( E, D1, D2 )

WRITE ( *, ’(F18.9)’ ) D1+D2

The output (i.e. given JD, Besselian epoch, equivalent JD, Julian epoch, equivalent JD) is:

2457073.05631

B2015.1365941021

2457073.056310000

J2015.1349933196

2457073.056310000

3 Time scales

3.1 Time in astronomy

Calculations in any scientific discipline may involve precise time, but what sets astronomy apart
is the number and variety of time scales that have to be used. There are several reasons for
this: astronomers must continue to deal with the very phenomena that lie behind obsolete
time scales, in particular the rotation of the Earth and the motions of the planets; as new
time scales have been introduced, continuity with the past has been preserved, leaving in the
various astronomical time scales a fossil record of former offsets and rates; and in astronomical
applications the physical context of the “clock” matters, whether it is on Earth, moving or
stationary, or on a spacecraft.

3.2 SOFA support for time scales

SOFA provides routines to handle the following seven time scales, all important to astronomers:

• TAI (International Atomic Time): the official timekeeping standard.

• UTC (Coordinated Universal Time): the basis of civil time.

• UT1 (Universal Time): based on Earth rotation.

• TT (Terrestrial Time): used for solar system ephemeris look-up.2

• TCG (Geocentric Coordinate Time): used for calculations centered on the Earth in space.

• TCB (Barycentric Coordinate Time): used for calculations beyond Earth orbit.

• TDB (Barycentric Dynamical Time): a scaled form of TCB that keeps in step with TT
on the average.

Note that the above list does not include:

• UT0, UT2: specialist forms of universal time that take into account polar motion and
known seasonal effects; no longer used.

• Sidereal time—which is an angle rather than a time.3

• GMT (Greenwich mean time): an obsolete time scale that can be taken to mean either
UTC or UT1.

• ET (ephemeris time): superseded by TT and TDB.

2Strictly TDB, but for most applications TT is more than good enough. See Section 3.6.
3The same can be said of UT1; however, the interrelation between UTC and UT1 makes it clearer and more

convenient to treat the latter as a time in the present document. Sidereal time is covered in the document SOFA
Tools for Earth Attitude.

• TDT (terrestrial dynamical time): the former name of TT.

• Civil time (BST, PDT etc.): these are UTC offset by some number of hours (or half hours,
or even quarter hours, in some cases), depending on longitude and time of year.

• Solar (sundial) time.

• GPS time: strictly speaking not itself an officially recognized time scale, but an accurate
and inexpensive way of bringing precise time into an observatory (see Section 3.4).

Of the seven time scales to be described here, one is atomic time (TAI), one is solar time (UT1),
one is an atomic/solar hybrid (UTC) and four are dynamical times (TT, TCG, TCB, TDB). Each
has a distinct role, and there are offsets of tens of seconds between some of them: when planning
an astronomical calculation it is vital to choose the right one. A particularly common mistake is
to assume that there is just one sort of precise time, namely UTC, compatible with everything
from telescope pointing (which actually requires UT1) to looking up planetary positions (which
requires TDB, which may be approximated by TT). In fact UTC itself is almost never the time
scale to use for astronomical calculations, except perhaps for record-keeping.

Figure 1 is a “route map” showing how the seven SOFA-supported time scales are related,
and how a time on one scale can be transformed into the same time on another scale. The time
scales are represented by the rectangular boxes, and their relationships are shown in the rounded
boxes. n.b. Both the diagram and the text gloss over many fine details, in order to concentrate
on practical applications.

The transformation from one time scale to the next can take a number of forms. In some cases,
for example TAI to TT, it is simply a fixed offset. In others, for example TAI to UT1, it is
an offset that depends on observations and cannot be predicted in advance (or only partially).
Some time scales, for example TT and TCG, are linearly related, with a rate change as well as an
offset. Others, for example TCG and TCB, require a 4-dimensional spacetime transformation.

3.3 Relativistic considerations

Although the majority of astronomical calculations do not explicitly involve general relativity
(GR), and can pretend that all points in space share an absolute time, it is wise to be aware of
how GR underpins the scheme set out in Figure 1.

In the spacetime coordinates (t, x, y, z) of an event, the temporal coordinate, t, is the proper
time that would be read by a (perfect) clock at that location. But in some applications, such
as the analysis of pulsar observations, it is convenient for the proper time to be remapped into
a differently curved spacetime, in which analysis is simpler and physical interpretation more
direct: this introduces the so-called coordinate time scales TCG, TCB and TDB. Proper time is
in fact the exception rather than the rule: even TAI is, strictly speaking, a coordinate time scale,
and the proper time read by the observer’s clock must in principle undergo a preliminary site-
dependent GR transformation before the diagram can be applied. But for terrestrial observers
proper time can, in practice, be regarded as the same as TAI, true to picosecond accuracy in
the case of clocks at sea level.

Note that Figure 1 depicts the time of a unique event in a single place (not necessarily on the
Earth), expressed in different forms; dubious concepts such as “the time read by a clock at the
barycenter” play no part, nor are light-time effects taken into account. For any event, anywhere,
it is possible to assign a TAI, a TT, a TDB and so on.

3.4 Atomic time: TAI

The unit of proper time is the SI second, defined as “the duration of 9,192,631,770 periods of the
radiation corresponding to the transition between the two hyperfine levels of the ground state
of the caesium 133 atom”, the latter being at 0 K and in zero magnetic field. The duration was
chosen to match the existing astronomical time scale, and is consequently a fossil of the solar
day length during the 19th century. The SI second is the unit of TAI and UTC, and is inherited
by the relativistic time scales TCG and TCB.4

TAI is a laboratory time scale, independent of astronomical phenomena apart from having been
synchronized to solar time when first introduced (at the start of 1958). It is realized through
a changing population of about 200 high-precision atomic clocks held at standards institutes
in various countries. There is an elaborate process of continuous intercomparison, leading to a
weighted average of all the clocks involved. TAI is close to proper time for an observer on the
geoid, and is an appropriate choice for terrestrial applications where continuity through UTC
leap seconds (see 3.5.1) is a requirement. It is not disseminated directly as a time service, but
can easily be realized from GPS or UTC.5 The SOFA software supports transformations between
TAI and three other time scales, namely UTC, UT1 and TT.

3.5 Solar Time: UT1 and UTC

UT1 (or plain UT) is the modern equivalent of mean solar time, and is really an angle rather than
time in the physics sense. Originally defined in terms of a point in the sky called “the fictitious
mean Sun”, UT1 is now defined through its relationship with Earth rotation angle (formerly
through sidereal time). Because the Earth’s rotation rate is slightly irregular—for geophysical
reasons—and is gradually decreasing, the UT1 second is not precisely matched to the SI second.
This makes UT1 itself unsuitable for use as a time scale in physics applications. However, some
applications do require UT1, such as pointing a telescope or antenna at a celestial target, delay
calculations in interferometers, and diurnal aberration, parallax and Doppler corrections.

UTC is a compromise between the demands of precise timekeeping and the desire to maintain
the current relationship between civil time and daylight. Since its introduction in 1960, UTC
has been kept roughly in step with UT1 by a variety of adjustments agreed in advance and

4The other two relativistic time scales, TT and TDB, can be regarded either as coordinate times in their own
right or as TCG and TCB expressed in different units.

5The Global Positioning System (GPS) maintains its own atomic time scale, steered to track UTC without
any adjustment for leap seconds implemented after 1980. Because GPS time was set to match UTC in 1980, it
inherited the TAI−UTC value from that era. Consequently, GPS time remains forever 19s behind TAI, to sub-
microsecond accuracy. The main reasons SOFA does not provide support for GPS time are (i) the transformation,
namely a fixed 19s offset, is trivial, and (ii) the danger of confusion between GPS system time itself and the time
displayed by most GPS receivers, which typically is UTC.

then carried out in a coordinated manner by the providers of time services—hence the name.
Though rate changes were used until 1972, since then all such adjustments have been made by
occasionally inserting a whole second, called a leap second, a procedure that can be thought of
as stopping the UTC clock for a second to let the Earth catch up. Leap seconds are discussed
below.

To obtain UT1 starting from UTC, it is necessary to look up the value of ∆UT1 = UT1−UTC for
the date concerned in tables published by the International Earth Rotation and Reference Sys-
tems Service (IERS); this is then added to the UTC. The quantity UT1−UTC, which typically
changes by 1-2 ms per day, can be obtained only by observation, principally very long baseline
interferometry (VLBI) using extragalactic radio sources, though seasonal effects are present and
the IERS listings are able to predict some way into the future with adequate accuracy for most
applications.

There are SOFA routines to link UTC, UT1, TAI and TT.

3.5.1 Leap seconds

Note that in Figure 1 there are two ways of getting from TAI to UT1, one via UTC and the
other directly. In a practical application (such as a telescope control system), the direct route
has the important advantage that the offset varies in a continuous manner, whereas the route
via UTC introduces the complication of dealing with leap seconds.

At the time of writing, leap seconds are introduced as necessary to keep UT1−UTC in the
range ±0s.9. The decision is made by international agreement, and announced several months in
advance by the IERS. Leap seconds occur usually at the end of December or June. Because on the
average the solar day is now 1-2ms longer than the nominal 86,400 SI seconds, accumulating to 1s

over a period of 18months to a few years, leap seconds are in practice always positive; however,
provision exists for negative leap seconds if needed. Each time a leap second is introduced,
the offset ∆AT = TAI−UTC changes by exactly 1s. Up-to-date information on TAI−UTC is
available from the IERS.

The form of a leap second can be seen from the following description of the end of 2008:

UTC ∆UT1 ∆AT UT1

2008 Dec 31 23 59 58 −0.593 +33 23 59 57.407
23 59 59 −0.593 +33 23 59 58.407
23 59 60 −0.593 +33 23 59 59.407

2009 Jan 1 00 00 00 +0.407 +34 00 00 00.407
00 00 01 +0.407 +34 00 00 01.407

Note that UTC has to be expressed as hours, minutes and seconds (or at least in seconds in a
given day) if leap seconds are to be taken into account in the correct manner. In particular, it
is inappropriate to express UTC as a Julian Date, because there will be an ambiguity during a

leap second—so that for example 1994 June 30 23h 59m 60s.0 and 1994 July 1 00h 00m 00s.0 would
both come out as MJD 49534.00000—and because subtracting two such JDs would not yield the
correct interval in cases that contain leap seconds.

Leap seconds pose tricky problems for software writers, and consequently there are concerns that
these events put safety-critical systems at risk. The correct solution is for designers to base such
systems on TAI or some other glitch-free time scale, not UTC, but this option is often overlooked
until it is too late. As the Earth rotation slows, leap seconds will become ever more frequent,
increasing the administrative burden and multiplying the danger of disruption. This has led
to proposals to relax the ±0s.9 requirement for UT1−UTC, putting off further adjustments for
several hundred years, at which point they would amount to timezone changes. The main
danger in ceasing leap seconds is that there are in existence unknown numbers of applications
that rely on UTC being a reasonable approximation to UT1, and these will gradually fail as
the discrepancy grows. However, because it is very likely that leap seconds will soon cease, it
is essential that all new applications, however modest their accuracy needs, take UT1−UTC
properly into account.

SOFA support for leap seconds is built into the various transformations that link UTC with TAI
and UT1. These call a SOFA routine that contains the historical list of leap seconds, namely
DAT (see Section 4.5.)

3.5.2 Delta T

The difference between UT1 and TT (formerly ET) is called ∆T, and in the present era can be
written out as

∆T = TT−UT1 = 32s.184 + ∆AT−∆UT1.

∆T is important for interpreting historical observations of solar system phenomena, where mod-
ern ephemerides provide the time of the event accurately but its appearance at a specified
geographical location depends on knowing the Earth orientation.

Because the Earth’s rotation is slowing due to tidal friction, and the rotation rate decreases
approximately linearly with time, ∆T increases quadratically. Empirical models for ∆T exist,
based on historical records such as eclipse sightings, but there is at present no SOFA support
in this area. However, SOFA routines are provided to transform between UT1 and TT given a
user-supplied ∆T value.

3.6 The dynamical time scales: TT, TCG, TCB and TDB

TT is a successor to the former Ephemeris Time, and is what is in practice6 used to look up
solar system ephemerides. The coordinate time scales TCG, TCB and TDB are the independent
variable in GR-based theories which describe the motions of bodies in the vicinity of the Earth

6TDB or a specific time scale belonging to the ephemeris concerned (denoted Teph) should really be used.
However, going to the trouble of computing TDB or Teph is usually not justified: even for the Moon, the error
from using TT instead of TDB is less than 1 milliarcsecond.

(TT, TCG) and in the solar system (TCB, TDB). Of the four, most observational astronomers
will need only TT, and perhaps TDB.

TT and TDB are close to each other (less than 2ms) and run at the same rate as TAI (exactly in
the case of TT; on the average in the case of TDB): they are the modern equivalents of ET. TCG
and TCB, used in theoretical work, run at different rates and so have long term drifts relative to
TAI. The four time scales are tied to TAI though the spacetime event 1997 January 1st 0h TAI
at the geocenter. The times of that event are as follows:

TAI 1997 Jan 1 00:00:00
TCG 1997 Jan 1 00:00:00 + 32.184 s
TCB 1997 Jan 1 00:00:00 + 32.184 s
TDB 1997 Jan 1 00:00:00 + 32.184 s − 65.5µs

3.6.1 Using solar system ephemerides: a warning

TT (or more correctly TDB) is the time scale to use when interrogating an ephemeris such as
JPL DE405, or when using published formulae to predict the position of the Earth or another
solar system body. It is an extremely common mistake to use UTC, in which case the
results will (at the time of writing) be well over a minute out. (The other common blunder is
to use the observer’s time rather than when the light left the source.)

3.6.2 TT

Terrestrial time, TT (called TDT between 1984 and 2000), is the theoretical time scale for clocks
at sea-level: for practical purposes it is tied to TAI through the fixed formula:7

TT = TAI + 32.184 s.

Calculating TT from UTC, rather than TAI, requires leap seconds to be taken into account: the
quantity ∆TT = TT−UTC for the UTC in question can be deduced from information provided
by the IERS. (See also the example in Section 1.5.)

SOFA routines are provided linking TT with TAI, TCG, TDB and UT1.

3.6.3 GCRS and BCRS, and spacetime units

TCG and TCB are the time coordinates of two IAU spacetime metrics called, respectively, the
geocentric and barycentric celestial reference systems (GCRS and BCRS). Dynamics problems

7The fixed 32s.184 offset reflects the history of trying to tie together clock timekeeping with Earth rotation and
lunar ephemerides. It retains continuity with the former ET, and represents the drift in mean solar time between
the ephemerides constructed in the 19th century and the inception of atomic time in the 1950s.

evaluated in these reference systems use physical constants unchanged from their familiar labo-
ratory values. When TT or TDB are to be used for the time coordinate, it is possible to apply
the appropriate scaling factor to the entire metric tensor, to preserve the standard relationship
c = 299 792 458ms−1. Under these circumstances, physical values such as mass parameters have
to be scaled as well. To avoid confusion over units, it is best to embrace the view that the orig-
inal proper units, such as SI seconds and metres, still apply, projected into the new metric, and
never to attach adjectives to units themselves. For example, we might say “The TDB interval
is 21.552044 s.” or “The TDB-compatible value for GME is 3.98004356× 1014m3s−2.”.

3.6.4 TCG and TCB

Geocentric coordinate time, TCG, is appropriate for theoretical studies of geocentric ephemerides.
Its relationship with TT is this conventional linear transformation:

TCG = TT+ LG × (JDTT − TT0),

where TT0 = 2443144.5003725 (i.e. TT at 1977 January 1.0 TAI), JDTT is TT expressed as
Julian date, and LG = 6.969290134× 10−10. The rate change LG means that TCG gains about
2s.2 per century with respect to TT or TAI; this represents the combined effect on the terrestrial
clock of the gravitational potential from the Earth and the observatory’s diurnal speed.

Barycentric coordinate time, TCB, is appropriate for applications where the observer is imagined
to be stationary in the solar system but with all the matter (in particular the Sun) absent. The
transformation from TCG to TCB thus takes account of the orbital speed of the geocenter and
the gravitational potential from the Sun and planets. The difference between TCG and TCB
involves a full 4-dimensional GR transformation:

TCB− TCG = c−2

{∫ t

t0

[
v2e
2

+ Uext(xe)]dt+ ve · (x− xe)

}
+O(c−4),

where the vectors xe and ve denote the barycentric position and velocity of the geocenter, the
vector x is the barycentric position of the observer and Uext is the Newtonian potential of all of
the solar system bodies apart from the Earth, evaluated at the geocenter. In this formula, t is
TCB and t0 is chosen to be consistent with 1977 January 1.0 TAI. The neglected terms, O(c−4),
are of order 10−16 in rate for terrestrial observers. Note that

(i) Uext(xe, xe) and ve are all ephemeris-dependent, and so the resulting TCB belongs to that
particular ephemeris, and

(ii) the term ve · (x− xe) is zero at the geocenter.

The linear relationship between TT and TCG is implemented in the SOFA routines TTTCG
and TCGTT. SOFA does not provide direct transformations between TT and TCB, instead
working via TDB (see the next Section).



3.6 The dynamical time scales: TT, TCG, TCB and TDB 17

3.6.5 TDB

The combined effect of the terrestrial observer’s orbital speed and the gravitational potential
from the Sun and planets is that a TCB clock gains on TT or TAI by nearly 0s.5 per year.
Barycentric dynamical time, TDB, suppresses this drift by applying a conventional linear trans-
formation:

TDB = TCB− LB × (JDTCB − T0)× 86400 + TDB0,

where JDTCB is TCB is expressed as Julian date, T0 = 2443144.5003725, LB = 1.550519768 ×
10−8 and TDB0 = −6.55 × 10−5 s. TDB thus functions as a coordinate time for barycentric
applications but, unlike TCB, stays close to TT on the average. The difference TDB−TT is
quasi-periodic, dominated by an annual term of amplitude 1.7 ms. For many purposes, the
difference is negligible and TT itself can be used instead of TDB. But for some applications—
an example being the long-term analysis of pulse arrival times from millisecond pulsars—the
difference cannot be neglected and use of TDB (or TCB) is essential.

When, as is usually the case, TT is available but TDB required, the full transformation chain
TT → TCG → TCB → TDB can be compressed into a single step. The resulting adjustment is
a quasi-periodic quantity8 dominated by an annual sinusoid of amplitude 1.7ms.

The linear relationship between TT and TDB is implemented in the SOFA routines TTTDB and
TDBTT. These routines require a user-supplied TDB−TT value, and the user is free to choose
whichever model or tabulation suits the application in hand. The SOFA routine DTDB provides
one such model, and is accurate to a few nanoseconds; see Section 4.3.4 for more information.

8The 1979 IAU resolution that introduced TDB stipulated that it differ from TDT by periodic terms only.
Although this describes the character of the TDB−TT quantity well, it proved impossible to pin down the
statement mathematically, and in 2006 the IAU simplified the definition of TDB by making it an ad hoc linear
transformation of TCB.

4 The SOFA time scale transformation routines

4.1 Architecture

SOFA provides routines that link adjacent pairs of time scales in Figure 1, leaving it to the user
to select those needed to construct the desired chain. This raises the obvious question:

Why isn’t there an all-purpose SOFA routine that simply accepts a time with respect
to one nominated scale and returns the time with respect to another nominated scale?

The answer is that some of the transformations involve supplementary quantities that can be
provided only by the application, either because they cannot be predicted or because they involve
choice. A general-purpose routine would need all these quantities as arguments, even though for
any given transformation (TAI to TT say) some or all would never be accessed. The example
in Section 4.4 gives an idea of what is involved: much of the clutter comes from the TT-to-
TDB transformation, which is a function of the observer’s position and involves a complicated
and computationally heavy ephemeris series. Although the user could simply supply all the
quantities every time, just in case, this would be a nuisance, and also computationally wasteful.

Of course, a knowledgeable user would know in advance which quantities are needed and could
supply harmless values (zeroes say) for all the others; but a user with that degree of expertise can
just as easily build the required chain—and may in any case like to have the intermediate results
available. Choosing which routines to call is made easy by Figure 2 which, like Figure 1, shows
the transformation chains, but also gives the names of the 16 routines and lists their arguments.
An application needing to transform UTC into TDB, for example, could first call UTCTAI, then
call TAITT, then TTTDB. Only the final call would involve an additional quantity, in this case
TDB−TT, which the user could obtain in a number of ways, most conveniently by calling the
DTDB routine.

4.2 Internal representation of times

SOFA has an existing convention for expressing times conveniently and to high precision, namely
two-part Julian date (see Section 2.3). The time transformation routines use this convention, so
that (for example) the TAI accepted by the TAITT routine is merely an ordinary SOFA two-
part JD. However, for reasons already described (Section 3.5.1), it is inappropriate to express
UTC as JD; if the straightforward CAL2JD, TF2D, JD2CAL and JDCALF routines are used
to transform between UTC and JD, there will be an ambiguity during a leap second, the leap
second and the subsequent second both returning the same range of JDs.

Rather than using a completely different representation—either for UTC as a special case or
for all the transformation routines—the SOFA designers decided instead in the case of UTC
to adapt the familiar two-part JD into a form that disguises the leap-second difficulties. This

name transformation arguments

TAIUTC TAI ⇒ UTC TAI1, TAI2, UTC1, UTC2

UTCTAI UTC ⇒ TAI UTC1, UTC2, TAI1, TAI2

UTCUT1 UTC ⇒ UT1 UTC1, UTC2, DUT, UT1, UT2

UT1UTC UT1 ⇒ UTC UT1, UT2, DUT, UTC1, UTC2

TAIUT1 TAI ⇒ UT1 TAI1, TAI2, DTA, UT1, UT2

UT1TAI UT1 ⇒ TAI UT1, UT2, DTA, TAI1, TAI2

TTUT1 TT ⇒ UT1 TT1, TT2, DT, UT1, UT2

UT1TT UT1 ⇒ TT UT1, UT2, DT, TT1, TT2

TAITT TAI ⇒ TT TAI1, TAI2, TT1, TT2

TTTAI TT ⇒ TAI TT1, TT2, TAI1, TAI2

TTTCG TT ⇒ TCG TT1, TT2, TCG1, TCG2

TCGTT TCG ⇒ TT TCG1, TCG2, TT1, TT2

TTTDB TT ⇒ TDB TT1, TT2, DTR, TDB1, TDB2

TDBTT TDB ⇒ TT TDB1, TDB2, DTR, TT1, TT2

TDBTCB TDB ⇒ TCB TDB1, TDB2, TCB1, TCB2

TCBTDB TCB ⇒ TDB TCB1, TCB2, TDB1, TDB2

Figure 2: The SOFA time scale transformation routines. Argument pairs TAI1,TAI2 etc.
are the encoded times produced by the DTF2D routine. DUT is UT1−UTC, DTA is UT1−TAI,
DT is TT−UT1, and DTR is TDB−TT, all in seconds. There is also in each case a final integer
argument for the returned status (0= success).

modification gives the various time transformation routines a consistent appearance, but at the
expense of introducing a non-standard internal scheme in the UTC case.9

Two dedicated routines, DTF2D and D2DTF, are available to handle the entire encode/decode
between formatted date and time—years, month, day, hour, minute and seconds—and two-part
JD (or, in the case of UTC, quasi-JD). The identifier for the time scale in question is one of the
arguments and, if it is ’UTC’, special treatment to deal with leap seconds is performed. It is
recommended that these routines are used for all date and time encode/decode operations, and
mandatory for UTC.

Here is example code that takes a time expressed as TAI, encodes it into the internal format
using the DTF2D routine, and transforms it into UTC; the TAI and UTC are each decoded
using the D2DTF routine and reported:

INTEGER J, IY, IM, ID, IHMSF(4)

DOUBLE PRECISION A1, A2, U1, U2

* Encode TAI date and time into internal format.

CALL iau_DTF2D ( ’TAI’, 2009, 1, 1, 0, 0, 33.7D0,

: A1, A2, J )

IF ( J.NE.0 ) STOP

* Decode and report the TAI.

CALL iau_D2DTF ( ’TAI’, 3, A1, A2, IY, IM, ID, IHMSF, J )

IF ( J.NE.0 ) STOP

WRITE ( *,

: ’(’’ TAI’’,I5,2(’’/’’,I2.2),I3,2(’’:’’,I2.2),’’.’’,I3.3)’)

: IY, IM, ID, IHMSF

* Transform TAI to UTC.

CALL iau_TAIUTC ( A1, A2, U1, U2, J )

IF ( J.NE.0 ) STOP

* Decode and report the UTC.

CALL iau_D2DTF ( ’UTC’, 3, U1, U2, IY, IM, ID, IHMSF, J )

IF ( J.NE.0 ) STOP

WRITE ( *,

: ’(’’ UTC’’,I5,2(’’/’’,I2.2),I3,2(’’:’’,I2.2),’’.’’,I3.3)’)

: IY, IM, ID, IHMSF

The output is:

9The quasi-JD that SOFA uses internally to represent UTC is obtained by treating days that end in a leap
second as having a different length from usual, namely 86401s (or 86399s in the case of a negative leap second),
when forming the fractional part. The scheme has no official status outside SOFA, and it is strongly recommended
that applications do not exploit it.



4.3 The supplementary quantities 21

TAI 2009/01/01 0:00:33.700

UTC 2008/12/31 23:59:60.700

Note that the test time is partway through a leap second.

4.3 The supplementary quantities

In Figure 2, while some of the transformations require only a time and the identities of the
“before” and “after” time scales, others require additional information that for one reason or
another can be supplied only by the application. In the latter cases, the required supplementary
quantity has either to be obtained from a service that monitors Earth rotation or involves a
choice of model.

4.3.1 UT1 minus UTC

In Figure 2, the argument DUT is the quantity ∆UT1 = UT1−UTC, in seconds. It can be
obtained from tables published by the IERS. For example, the following extract from one of the
IERS long-term tabulations covers the date used in the example in Section 4.4, later.

INTERNATIONAL EARTH ROTATION AND REFERENCE SYSTEMS SERVICE

EARTH ORIENTATION PARAMETERS

EOP (IERS) 05 C04

FORMAT(3(I4),I7,2(F11.6),2(F12.7),2(F11.6),2(F11.6),2(F11.7),2F12.6)

**********************************************************************************

Date MJD x y UT1-UTC LOD ...

" " s s ...

(0h UTC)

1962 1 1 37665 -0.012700 0.213000 0.0326338 0.0017230 ...

1962 1 2 37666 -0.015900 0.214100 0.0320547 0.0016690 ...

1962 1 3 37667 -0.019000 0.215200 0.0315526 0.0015820 ...

:

2006 1 13 53748 0.048417 0.380702 0.3337689 -0.0002682 ...

2006 1 14 53749 0.048576 0.380378 0.3339950 -0.0001443 ...

2006 1 15 53750 0.049116 0.380121 0.3340826 -0.0000172 ...

2006 1 16 53751 0.049864 0.380044 0.3340667 0.0000669 ...

2006 1 17 53752 0.050329 0.380310 0.3339361 0.0002053 ...

:

4.3.2 UT1 minus TAI

In Figure 2, the argument DTA is the quantity UT1−TAI, in seconds. It can be obtained by
calling the SOFA routine DAT (see Section 4.5) to obtain ∆AT, getting ∆UT1 from IERS tables,
and subtracting:

UT1−TAI = ∆UT1−∆AT.

4.3.3 TT minus UT1

In Figure 2, the argument DT is the quantity ∆T = TT−UT1, in seconds. For current dates it
can be constructed accurately by knowing ∆UT1 and ∆AT and using:

∆T = 32s.184 + ∆AT−∆UT1.

However, ∆T is more commonly used in applications dealing with observations made many
decades or centuries ago, and a suitable value must be chosen by the user, either from tables,
for example in The Astronomical Almanac, or using a model. (See also Section 3.5.2.)

4.3.4 TDB minus TT

In Figure 2, the argument DTR is the quantity TDB−TT, in seconds. The quantity depends on
the adopted solar system ephemeris, and is normally obtained either by evaluating a series or
by interrogating a precomputed time ephemeris. (See also Section 3.6.5.)

The SOFA routine DTDB provides a detailed model (nearly 800 terms) for TDB−TT, accurate
to a few nanoseconds in the present era when compared with the latest work. Its arguments are
TDB (TT can be used instead without materially affecting the results), UT1, and the observer’s
coordinates, expressed as longitude and the distances from the rotation axis and equator. If the
application can tolerate errors of up to 2µs, zeroes can be used for the final four arguments,
giving a geocentric result. The example in the next Section demonstrates use of the DTDB
routine, including what is involved in transforming the observer’s terrestrial coordinates into
the required form.

For applications where the distinction between TT and TDB matters but 50µs accuracy is
sufficient, and the date lies in the range 1980-2100, the following simplified expression can be
used:

TDB ≃ TT+ 0s.001657 sin g

where g = 6.24 + 0.017202 × (JDTT − 2451545) approximates the Earth’s mean anomaly in
radians.

4.4 A comprehensive example: transform UTC into other times

In the following example, an observer at north latitude +19◦ 28′ 52′′.5, west longitude 155◦ 55′ 59′′.6,
at sea level, on 2006 January 15 at 21:24:37.5 UTC, requires the time in all the other supported
time scales:

INTEGER LATND, LATNM, LONWD, LONWM, J, IY, MO, ID, IH, IM,

: IHMSF(4)

DOUBLE PRECISION SLATN, SLONW, HM, ELON, PHI, XYZ(3), U, V,

: SEC, UTC1, UTC2, DUT, UT11, UT12, UT,

: TAI1, TAI2, TT1, TT2, TCG1, TCG2, DTR,

: TDB1, TDB2, TCB1, TCB2

DOUBLE PRECISION iau_DTDB

* Site terrestrial coordinates (WGS84).

LATND = 19

LATNM = 28

SLATN = 52.5D0

LONWD = 155

LONWM = 55

SLONW = 59.6

HM = 0D0

* Transform to geocentric.

CALL iau_AF2A ( ’+’, LATND, LATNM, SLATN, PHI, J )

IF ( J.NE.0 ) STOP

CALL iau_AF2A ( ’-’, LONWD, LONWM, SLONW, ELON, J )

IF ( J.NE.0 ) STOP

CALL iau_GD2GC ( 1, ELON, PHI, HM, XYZ, J )

IF ( J.NE.0 ) STOP

U = SQRT ( XYZ(1)*XYZ(1) + XYZ(2)*XYZ(2) )

V = XYZ(3)

* UTC date and time.

IY = 2006

MO = 1

ID = 15

IH = 21

IM = 24

SEC = 37.5D0

* Transform into internal format.

CALL iau_DTF2D ( ’UTC’, IY, MO, ID, IH, IM, SEC, UTC1, UTC2, J )

IF ( J.NE.0 ) STOP

* UT1-UTC (s, from IERS).

DUT = +0.3341D0

* UTC -> UT1.

CALL iau_UTCUT1 ( UTC1, UTC2, DUT, UT11, UT12, J )

IF ( J.NE.0 ) STOP

* Extract fraction for TDB-TT calculation, later.

UT = MOD ( MOD(UT11,1D0)+MOD(UT12,1D0), 1D0 )

* UTC -> TAI -> TT -> TCG.

CALL iau_UTCTAI ( UTC1, UTC2, TAI1, TAI2, J )

IF ( J.NE.0 ) STOP

CALL iau_TAITT ( TAI1, TAI2, TT1, TT2, J )

IF ( J.NE.0 ) STOP

CALL iau_TTTCG ( TT1, TT2, TCG1, TCG2, J )

IF ( J.NE.0 ) STOP

* TDB-TT (using TT as a substitute for TDB).

DTR = iau_DTDB ( TT1, TT2, UT, ELON, U, V )

* TT -> TDB -> TCB.

CALL iau_TTTDB ( TT1, TT2, DTR, TDB1, TDB2, J )

IF ( J.NE.0 ) STOP

CALL iau_TDBTCB ( TDB1, TDB2, TCB1, TCB2, J )

IF ( J.NE.0 ) STOP

* Report.

CALL iau_D2DTF ( ’UTC’, 6, UTC1, UTC2, IY, MO, ID, IHMSF, J )

IF ( J.NE.0 ) STOP

WRITE ( *, 1 ) ’UTC’, IY, MO, ID, IHMSF

CALL iau_D2DTF ( ’UT1’, 6, UT11, UT12, IY, MO, ID, IHMSF, J )

IF ( J.NE.0 ) STOP

WRITE ( *, 1 ) ’UT1’, IY, MO, ID, IHMSF

CALL iau_D2DTF ( ’TAI’, 6, TAI1, TAI2, IY, MO, ID, IHMSF, J )

IF ( J.NE.0 ) STOP

WRITE ( *, 1 ) ’TAI’, IY, MO, ID, IHMSF

CALL iau_D2DTF ( ’TT’, 6, TT1, TT2, IY, MO, ID, IHMSF, J )

IF ( J.NE.0 ) STOP

WRITE ( *, 1 ) ’TT ’, IY, MO, ID, IHMSF

CALL iau_D2DTF ( ’TCG’, 6, TCG1, TCG2, IY, MO, ID, IHMSF, J )

IF ( J.NE.0 ) STOP

WRITE ( *, 1 ) ’TCG’, IY, MO, ID, IHMSF

CALL iau_D2DTF ( ’TDB’, 6, TDB1, TDB2, IY, MO, ID, IHMSF, J )

IF ( J.NE.0 ) STOP

WRITE ( *, 1 ) ’TDB’, IY, MO, ID, IHMSF

CALL iau_D2DTF ( ’TCB’, 6, TCB1, TCB2, IY, MO, ID, IHMSF, J )

IF ( J.NE.0 ) STOP

WRITE ( *, 1 ) ’TCB’, IY, MO, ID, IHMSF

1 FORMAT ( 1X,A,I5,2(’/’,I2.2),I3,2(’:’,I2.2),’.’,I6.6 )

The output is:

UTC 2006/01/15 21:24:37.500000

UT1 2006/01/15 21:24:37.834100

TAI 2006/01/15 21:25:10.500000

TT 2006/01/15 21:25:42.684000

TCG 2006/01/15 21:25:43.322690

TDB 2006/01/15 21:25:42.683799

TCB 2006/01/15 21:25:56.893378

The UT1 might be used for pointing a telescope, the TT for looking up a planet’s position in
an ephemeris, and the TDB for interpreting pulsar observations. TAI would be appropriate
for calculating long intervals between observed events, as well as the best way of making time-
critical applications leap-second-proof. TCG, TCB would have more specialized dynamical uses.
The UTC itself would be suitable for logging and for conversion into local time, and essentially
nothing else.

4.5 The DAT routine

Applications that require explicit access to leap second information can call the DAT routine.
This contains a full list of leap seconds since their inception in 1972, as well as the rate changes
used before that. Here is how to obtain TAI−UTC for a given date:

INTEGER J

DOUBLE PRECISION DELTAT

* TAI-UTC for 0h UTC on 2009 Feb 13.

CALL iau_DAT ( 2009, 2, 13, 0D0, DELTAT, J )

IF ( J.NE.0 ) STOP

WRITE ( *, ’(SP,F6.1)’ ) DELTAT

The output is “+34.0”.

Note that because the DAT routine has to be updated each time a new leap second is announced,
there will inevitably be out-of-date copies in circulation, both in source form and inside object
libraries. Applications that call DAT must therefore be relinked occasionally, and moreover using
object libraries known to have been rebuilt from up-to-date SOFA source code. Rudimentary
checks inside the DAT routine do at least protect against too large a gap between the nominated
date and when the DAT source code was last changed. However, a better solution for operational
systems is to develop a local file-based method, eliminating the need to call DAT. When this
is done, DAT can still play an important role during development, as a source of test data.
If such an application uses the SOFA time transformation routines, which themselves call the
DAT routine, its own file-based capabilities can be used to recognize when relinking has become
necessary.

5 Further reading

Additional explanatory material can be found in The Astronomical Almanac and similar national
publications. An extensive glossary is available as part of The Astronomical Almanac Online:
see http://asa.usno.navy.mil/SecM/Glossary.html.

Detailed information on models and procedures can be found in:

• IERS Conventions (2003), D.D.McCarthy & G.Petit (eds.), Verlag des Bundesamts für
Kartographie und Geodäsie, Frankfurt am Main (2004),

and in successive updates. The following may also be useful for background reading:

• Explanatory Supplement to the Astronomical Almanac, ed. P.Kenneth Seidelmann (1992),
University Science Books.

See Section 2.553 of the book for a discussion of ∆T and references to the Stephenson & Morrison
and McCarthy & Babcock papers mentioned earlier.

